

スキルシミュレータメーカーのチュートリアル

2 版. 2021 年 5 月 11 日

5ch スキルシミュレータ開発 Ver.13 の 480

目次

1	はじめに	2
2	最初のシミュ	2
2.1	入力ファイルの内容の説明	3
3	スキルとスキル条件の指定	5
3.1	入力ファイルの内容の説明	6
4	装備とスキルの種類が増えた場合の工夫	7
4.1	入力ファイルの内容の説明	8
5	スロットと珠	10
5.1	入力ファイルの内容の説明	11
6	大きさの違うスロット	13
6.1	入力ファイルの内容の説明	14
7	スキルレベル上限解放、結果表示の工夫、追加スキル検索	15
7.1	入力ファイルの内容の説明	17
8	雑多な補足	20
8.1	ワンセット防具	20
8.2	検索結果が少ないこと（線形計画法）	20
8.3	追加検索の条件	21
8.4	パラメータ保存のタイミング	21
9	おわりに	21

1 はじめに

まず、1 版から 2 版に改訂した際に、入力ファイルの文法が大幅に変更され、互換性がなくなりましたので、ご注意下さい。

この文書はいわゆる「スキルシミュメーカー」^{*1}の利用方法のチュートリアルです。

- C や javascript のプログラムは不要です。
- 独自言語でスキルシミュレータを記述します。
- シミュレータの生成には、ruby の実行環境が必要ですが、ウェブ上で生成できるページも用意してあります (<http://nap.s3.xrea.com/singen-demo.html>)。このデモは、入力ファイルサイズが大きいと動作しないかも知れません^{*2}。
- モンスターハンターライズ専用ではないので、過去作品、将来の作品どころか、別のゲームにも使えるかも知れません。

このチュートリアルでは、シミュレータを生成するための独自言語の完全な仕様を説明することはせずに、ごく簡単なシミュレータを生成するための入力ファイル (18 行) から始めて、徐々に複雑なもの (最後は 75 行) の例をもとに説明をします。長くなりますが各段階で入力ファイル全文を掲載し、説明は前段階にはなかった部分について行います。

シミュレータは線形計画法を用いたもので、その原理を知りたい方は、<https://github.com/13-480/lp-doc/blob/main/lpsim-v4.pdf> を参照して下さい。よく見るシミュとは動作が違うのでこのチュートリアルの最後の方で補足します。

2 最初のシミュ

装備は、「フク」が 2 種類だけの簡単なスキルシミュレータの例を考えます。スキル、珠、スロットはまだありません。

フク	防御力
ポロシャツ	1
パーカー	2

入力ファイルは以下ようになります。その後に内容の説明をします。

```
# チュートリアルその 1
<META>
title チュートリアル 1 (最初のシミュ)

<UI>
```

^{*1} スキルシミュツールでも良いですが、ツールが KADOKAWA の商標なのでやめました。

^{*2} サイズの目安としては、モンスターハンターワールド：アイスボーンの装備やスキルの数であれば、入力ファイルサイズは 1 万行を超えます。プログラム不要といっても、この規模のテキストを手で作成することは現実的ではないので、何らかのプログラムで出力することにはなると思います。それでも、1 からシミュレータを作るよりは格段に易しいプログラムで済みます。

```

(0..C) ポロシャツ -> 1 防御力
(0..C) パーカー   -> 2 防御力

<RELATION>
ポロシャツ + パーカー <= 1

<QUERY>
query 検索 検索中止 追加検索 防御力

<SUMMARY>
v 防御力
n* ポロシャツ
n* パーカー

```

2.1 入力ファイルの内容の説明

■コメント

```
# チュートリアルその1
```

「#」以降行末まではコメントで、無視されます。


■META セクション

```

<META>
title チュートリアル1 (最初のシミュ)

```

title で始まる行で、生成されるウェブページのタイトルを指定します。



■UI セクション


```

<UI>
(0..C) ポロシャツ -> 1 防御力
(0..C) パーカー   -> 2 防御力

```

各装備の所持数や、スキル、スロット、属性耐性等のパラメータをどれだけ発動するかを指定します。

この例の「(0..C)」の部分は「ポロシャツ」あるいは「パーカー」の所持数の指定です。このような「(最小値.. 最大値)」という記述は、その装備をいくつ装備できるかの最小値・最大値を指定します。最大値の「C」はシミュレータ上に表示されるチェックボックスで指定されるという意味です。チェックボックスはチェックされていれば1、いなければ0を表します。



また、「-> 1 防御力」の部分は、ポロシャツを1つ装備すると、スキル等がどれだけ発動するかを指定しています。この例では、ポロシャツの防御力が1、パーカーの防御力が2と指定しています*3。

■RELATION セクション

```
<RELATION>
ポロシャツ + パーカー <= 1
```


装備の数などの満たすべき関係式を指定します。この例では、ポロシャツの装備数とパーカーの装備数の和が1以下であるという指定です。ポロシャツとパーカーのどちらか一方だけ1であるか、両方0だけが可能なので、装備できるのはどちらか一方、あるいは装備なしだけが可能であることを意味しています。

■QUERY セクション

```
<QUERY>
query 検索 検索中止 追加検索 防御力
```

検索ボタンは状態によって表示されるテキストが変わるので、それを指定します。query に続いて4つある文字列の最初の3つは、順に、初期状態の表示文字列、検索中の表示文字列、検索が終わって検索結果が画面に残っているときの表示文字列です。

最後の文字列は、何を最大化して検索するかを指定します。つまり、条件を満たすものたちのうち、防御力最大のものを検索するという指定です。



■SUMMARY セクション

*3 この例では、実際には、「**防御力 = ポロシャツ + 2 パーカー**」という式を、RELATION セクションに記述するのと同じです。RELATION セクションだけで済むのに UI セクションでも指定できるようにする理由は、現在2種類であるフクの種類が大量になった場合、この式が長大になってしまうので、これを避けたいからです。


```
<SUMMARY>
v 防御力
n* ポロシャツ
n* パーカー
```

検索結果として何を表示するか指定します。正確には、検索結果の「見出し」部分の表示です。

この例では、防御力、ポロシャツ、パーカーについて、合計 3 件の情報を表示します。行頭の記号の意味は以下のとおりです。

記号	n	v	*
意味	名前を表示	値を表示	値が 0 なら何も表示しない

従って、防御力はその値だけを表示し、ポロシャツとパーカーは、0 でなければ (UI セクションでの指定により 1 であることが確定) 名前だけ表示します。



3 スキルとスキル条件の指定

前節ではスキルがありませんでしたが、「攻撃」スキルを導入してみます。珠、スロットはまだありません。

フク	防御力	攻撃スキル
ポロシャツ	1	1
パーカー	2	0

入力ファイルの全体は以下のとおりで、前節からの変更箇所を赤くしています。入力ファイルの後に記す内容の説明は、変更のあった場所に限定します。

```
# チュートリアルその 2
<META>
title チュートリアル 2 (スキルとスキル条件の指定)

<UI>
(0..C) ポロシャツ -> 1 防御力, 1 攻撃
(0..C) パーカー   -> 2 防御力
br
```

```

([0 1 2 3 4 5]..5) 攻撃
(0?..) 防御力

<RELATION>
ポロシャツ + パーカー <= 1

<QUERY>
query 検索 検索中止 追加検索 防御力

<SUMMARY>
v 防御力
n* ポロシャツ
n* パーカー

<DETAILS>
newcolumn スキル
nv* 攻撃

```

3.1 入力ファイルの内容の説明

■UI セクション

```

<UI>
(0..C) ポロシャツ -> 1 防御力, 1 攻撃
(0..C) パーカー -> 2 防御力
br
([0 1 2 3 4 5]..5) 攻撃
(0?..) 防御力


```

ポロシャツが攻撃 Lv1 を発動するので、ポロシャツの行に攻撃スキルの指定を追加します。このように、複数のものを発動するときは、カンマ区切りで記述します。

「[UI] br」は改行です。次の UI 部品は、新しい行から表示されます。

「([0 1 2 3 4 5]..5) 攻撃」の最小値の部分は、0 から 5 までを選択できるプルダウンメニューを表しています。ポロシャツと同様の指定ですが、プルダウンが最小値の方に付きます。これにより、「攻撃スキルが 2 以上」などの条件指定が可能になります。ただし、シミュレータでの表示だけを見ても、最小値の指定か最大値の指定かは区別できませんので、文脈でわかるように UI 部品を配置する必要があります。

「(0?..) 防御力」の最小値の部分のように、数値の後に「?」を付けると、値をテキストボックスで入力できるようになります。この場合の「0」はテキストボックスの初期値です。また、最大値の記載がありませんが、その場合は上限なしとなります。




■DETAILS セクション

```
<DETAILS>
newcolumn スキル
nv* 攻撃
```

前節の例の検索結果表示は、検索結果の「見出し」部分の表示を指定する SUMMARY セクションのみでした。「見出し」部分をクリックすると、折りたたまれている「詳細」部分が現れますが、DETAILS セクションでは、「詳細」部分の表示を指定します。

詳細部分は、段組表示が基本で、「newcolumn スキル」は新しい段の開始で、1 行目に「スキル」の見出しを置きます。

「nv* 攻撃」は SUMMARY セクションと同じで、値が 0 でないときに限り、攻撃の名前と値を表示します。



4 装備とスキルの種類が増えた場合の工夫

前節では装備が 2 種類、スキルが 1 種類でしたが、これらを増やしてみます。種類が増えても、関係式が長くないような書き方、見辛くないような表示の体裁の指定の仕方を説明します。珠、スロットはまだありません。

		防御力	攻撃スキル	解放スキル
フク	ポロシャツ	1	1	0
	パーカー	2	0	0
クツ	サンダル	1	0	0
	スリッパ	2	0	1

入力ファイルの全体は以下のとおりで、前節からの変更箇所を赤くしています。入力ファイルの後に記す内容の説明は、変更のあった場所に限定します。

```
# チュートリアルその 3
<META>
title チュートリアル 3 (装備とスキルの種類が増えた場合の工夫)
```

```

<UI>
subsection 装備
width:150px
(0..1) フクなし:{フク}
(0..C) ポロシャツ:{フク} -> 1 防御力, 1 攻撃
(0..C) パーカー:{フク} -> 2 防御力
br
(0..1) クツなし:{クツ}
(0..C) サンドル:{クツ} -> 1 防御力
(0..C) スリッパ:{クツ} -> 2 防御力, 1 解放

subsection スキル条件・最低防御力指定
([0 1 2 3 4 5]..5) 攻撃:{スキル}
([0 1 2 3]..3)      解放:{スキル}
(0?..) 防御力

<RELATION>
{フク}.sum = {クツ}.sum = 1

<QUERY>
query 検索 検索中止 追加検索 防御力

<SUMMARY>
width:50px
v 防御力
width:150px
n* {フク}
n* {クツ}

<DETAILS>
newcolumn スキル
nv* {スキル}

```

4.1 入力ファイルの内容の説明

■UI、RELATION セクション

```

<UI>
subsection 装備
width:150px
(0..1) フクなし:{フク}
(0..C) ポロシャツ:{フク} -> 1 防御力, 1 攻撃
(0..C) パーカー:{フク} -> 2 防御力
br
(0..1) クツなし:{クツ}
(0..C) サンドル:{クツ} -> 1 防御力
(0..C) スリッパ:{クツ} -> 2 防御力, 1 解放

subsection スキル条件・最低防御力指定
([0 1 2 3 4 5]..5) 攻撃:{スキル}
([0 1 2 3]..3)      解放:{スキル}
(0?..) 防御力


<RELATION>
{フク}.sum = {クツ}.sum = 1

```


サンダルとスリッパが増えた分と、スリッパに付いている解放スキルの分を追記してあります。

「subsection 装備」は、「装備」を見出しにした、折りたたみ可能な部品で出力する指定です。装備が大量になったときは、必要なもの以外は折りたたむことを可能にします。

「width:150px」のように、UI 部品の幅を指定できます。指定を解除するときは「nowidth」とします。これにより、UI 部品を揃えることができます。SUMMARY セクションでも利用できます。




「フクなし:{フク}」などが装備やスキルの後に付いているときは、その装備やスキルをグループに登録します。グループ「{フク}」には、フクなし、ポロシャツ、パーカーの3つが登録されています。

RELATION セクションの、「{フク}.sum」は、グループに登録されているものたちの総和を意味します。これにより、フクは、ポロシャツ、パーカー、フクなしのいずれか1つを装備することになります。フクなしは、装備なしのときにも表示させるために導入しています。

「(0..1) フクなし」は、最小値・最大値とも定数ですので、UI 部品が生成されません。つまり、RELATION セクションに「0 <= フクなし <= 1」と記述するのと同じです。

攻撃スキルの最大レベルは5にしていたのですが、新設の解放スキルの最大は3としました。ともに、グループ「スキル」に登録しています。



チュートリアル3 (装備とスキルの種類が増えた場合の工夫)

▼ 装備

☒ ポロシャツ
☒ パーカー

☒ サンダル
☒ スリッパ

開いた状態

▼ スキル条件・最低防御力指定

1

攻撃

0

解放

0

防御力

検索

検索結果の全消去

■SUMMARY セクション

```
<SUMMARY>
width:50px
v 防御力
width:150px
n* {フク}
n* {クツ}
```

「n* { フク }」のようにグループを用いると、グループに登録されている、「n* フクなし」「n* ポロシャツ」「n* パーカー」の3つを指定したことになります。これは、DETAILS セクションでも同様です。

5 スロットと珠

防具に「スロット」があり、そこにスキルを持った「珠」を装着できることを考えてみます。このチュートリアルでは最終的には大・中・小3種類のスロットを考えますが、この節では小スロットのみを考え、次節で3種類のスロットを考えることにします。以下、入力ファイルでは、小スロットを短く「小スロ」と表します。

		防御力	攻撃スキル	解放スキル	小スロット
フク	ポロシャツ	1	1	0	0
	パーカー	2	0	0	2
クツ	サンダル	1	0	0	0
	スリッパ	2	0	1	1

珠	スキル	必要スロット
攻撃珠	攻撃 1	小

入力ファイルの全体は以下のとおりで、前節からの変更箇所を赤くしています。入力ファイルの後に記す内容の説明は、変更のあった場所に限定します。

```
# チュートリアルその 4
<META>
```

```

title チュートリアル 4 (スロットと珠)

<UI>
subsection 装備
width:150px
(0..1) フクなし:{フク}
(0..C) ボロシャツ:{フク} -> 1 防御力, 1 攻撃
(0..C) パーカー:{フク} -> 2 防御力, 2 小スロ
br
(0..1) クツなし:{クツ}
(0..C) サンドル:{クツ} -> 1 防御力
(0..C) スリッパ:{クツ} -> 2 防御力, 1 解放, 1 小スロ
br
(0..[0 1 2 3 4 5]) 攻撃珠:{珠} -> 1 攻撃, -1 小スロ

subsection スキル条件・最低防御力指定
([0 1 2 3 4 5]..5) 攻撃:{スキル}
([0 1 2 3]..3) 解放:{スキル}
(0?..) 防御力

<RELATION>
{フク}.sum = {クツ}.sum = 1
小スロ >= 0

<QUERY>
query 検索 検索中止 追加検索 防御力

<SUMMARY>
width:50px
v 防御力
width:150px
n* {フク}
n* {クツ}

<DETAILS>
newcolumn スキル
nv* {スキル}

newcolumn 珠
nv* {珠}

newcolumn 空きスロット
nv 小スロ

```

5.1 入力ファイルの内容の説明

■UI セクション

```

<UI>
subsection 装備
width:150px
(0..1) フクなし:{フク}
(0..C) ボロシャツ:{フク} -> 1 防御力, 1 攻撃
(0..C) パーカー:{フク} -> 2 防御力, 2 小スロ
br

```


```

(0..1) クツなし:{クツ}
(0..C) サンダル:{クツ} -> 1 防御力
(0..C) スリッパ:{クツ} -> 2 防御力, 1 解放, 1 小スロ
br
(0..[0 1 2 3 4 5]) 攻撃珠:{珠} -> 1 攻撃, -1 小スロ

```

パーカーが小スロットを2つ、スリッパが小スロットを1つ発動することを指定します。

また、攻撃珠が攻撃のスキルレベルを1発動し、小スロット1つを消費することも指定します。消費する場合は負の値になります。



■RELATION セクション

```

<RELATION>
{フク}.sum = {クツ}.sum = 1
小スロ >= 0

```

珠が小スロットに装着できるには、珠が必要とする小スロット数が、防具に備わっている小スロット数以下である必要があります。INDUCE セクションで、珠の消費する小スロット数を負で与えていますので、小スロの値は、消費されずに残っているスロット数です。従って、条件は「小スロ >= 0」と表せることになります。

■DETAILS セクション

```

<DETAILS>
newcolumn スキル
nv* {スキル}

newcolumn 珠
nv* {珠}

newcolumn 空きスロット
nv 小スロ

```

結果表示の詳細部分に、珠の個数と空きスロットを、ともに1列設けて表示します。

チュートリアル4 (スロットと珠)

▼ 装備

☒ ポロシャツ ☒ パーカー

☒ サンドル ☒ スリッパ

3 ☒ 攻撃珠

▼ スキル条件・最低防御力指定

1 ☒ 攻撃 解放 防御力

▶ 3 ポロシャツ スリッパ

▼ 3 パーカー サンドル

スキル 珠 空きスロット

攻撃 2 攻撃珠 2 小スロ 0

珠の列と空きスロットの列

▶ 4 パーカー スリッパ

6 大きさの違うスロット

前節では小スロットだけでしたが、この節では大・中・小3種類のスロットを考えます。鍵は、RELATIONセクションの関係式ですが、なぜそうなるかは省きます。<https://github.com/13-480/lp-doc/blob/main/lpsim-v4.pdf> に説明があります。

スキルシミュレーカの現在のバージョンでは、大・中・小のスロットそれぞれの空き数を得られず、大きさを見捨てた合計スロット数だけが得られます。スロットの空きが1次式で表せないのが理由で、現時点では入力ファイルを工夫するだけでは不可能です。

新たに、ポロシャツにスロットを4つ追加し、攻撃大珠も新設しました。

		防御力	攻撃スキル	解放スキル	小スロット	中スロット	大スロット
フク	ポロシャツ	1	1	0	2	1	1
	パーカー	2	0	0	2	0	0
クツ	サンダル	1	0	0	0	0	0
	スリッパ	2	0	1	1	0	0

珠	スキル	必要スロット
攻撃珠	攻撃 1	小
攻撃大珠	攻撃 2	大

入力ファイルの全体は以下のとおりで、前節からの変更箇所を赤くしています。入力ファイルの後に記す内容の説明は、変更のあった場所に限定します。

```

# チュートリアルその 5
<META>
title チュートリアル 5 (大きさの違うスロット)

<UI>
subsection 装備
width:150px
(0..1) フクなし:{フク}
(0..C) ボロシャツ:{フク} -> 1 防御力, 1 攻撃, 2 小スロ, 1 中スロ, 1 大スロ
(0..C) パーカー:{フク} -> 2 防御力, 2 小スロ
br
(0..1) クツなし:{クツ}
(0..C) サンドル:{クツ} -> 1 防御力
(0..C) スリッパ:{クツ} -> 2 防御力, 1 解放, 1 小スロ
br
(0..[0 1 2 3 4 5]) 攻撃珠:{珠} -> 1 攻撃, -1 小スロ
(0..[0 1 2]) 攻撃大珠:{珠} -> 2 攻撃, -1 大スロ

subsection スキル条件・最低防御力指定
([0 1 2 3 4 5]..5) 攻撃:{スキル}
([0 1 2 3]..3) 解放:{スキル}
(0?..) 防御力

<RELATION>
{フク}.sum = {クツ}.sum = 1

大スロ >= 0
大スロ + 中スロ >= 0
大スロ + 中スロ + 小スロ >= 0
空きスロット合計 = 大スロ + 中スロ + 小スロ

<QUERY>
query 検索 検索中止 追加検索 防御力

<SUMMARY>
width:50px
v 防御力
width:150px
n* {フク}
n* {クツ}

<DETAILS>
newcolumn スキル
nv* {スキル}

newcolumn 珠
nv* {珠}

newcolumn 空きスロット
nv 空きスロット合計

```

6.1 入力ファイルの内容の説明

■UI セクション

```

<UI>
subsection 装備
width:150px
(0..1) フクなし:{フク}
(0..C) ポロシャツ:{フク} -> 1 防御力, 1 攻撃, 2 小スロ, 1 中スロ, 1 大スロ
(0..C) パーカー:{フク} -> 2 防御力, 2 小スロ
br
(0..1) クツなし:{クツ}
(0..C) サンドル:{クツ} -> 1 防御力
(0..C) スリッパ:{クツ} -> 2 防御力, 1 解放, 1 小スロ
br
(0..[0 1 2 3 4 5]) 攻撃珠:{珠} -> 1 攻撃, -1 小スロ
(0..[0 1 2]) 攻撃大珠:{珠} -> 2 攻撃, -1 大スロ

```

ポロシャツにスロットを追加し、攻撃大珠を新設します。

■RELATION セクション

```

<RELATION>
{フク}.sum = {クツ}.sum = 1

大スロ >= 0
大スロ + 中スロ >= 0
大スロ + 中スロ + 小スロ >= 0
空きスロット合計 = 大スロ + 中スロ + 小スロ

```

スロットに関する関係式は、「小スロ >= 0」などとはならないです。このチュートリアルでは説明は省きます。ただし、「大スロ + 中スロ + 小スロ」は、空きスロット合計に一致するので、これだけは利用することにし、DETAILS セクションに表示します。

各スロットの空き数を求めることもできますが、次節で触れます。

7 スキルレベル上限解放、結果表示の工夫、追加スキル検索

ここでは、モンスターハンターワールド：アイスボーンで言うところの極意、つまり、スキルレベルの上限解放を説明します。

また、スキルの種類が増えた場合の、検索対象のスキルと、そうではないがおまけで付いてきたスキルを区別して表示する方法も説明します。

		防御力	攻撃スキル	解放スキル	小スロット	中スロット	大スロット
フク	ポロシャツ	1	1	0	2	1	1
	パーカー	2	0	0	2	0	0
クツ	サンダル	1	0	0	0	0	0
	スリッパ	2	0	1	1	0	0

珠	スキル	必要スロット
攻撃珠	攻撃 1	小
攻撃大珠	攻撃 2	大
解放珠	解放 1	中

入力ファイルの全体は以下のとおりで、前節からの変更箇所を赤くしています。入力ファイルの後に記す内容の説明は、変更のあった場所に限定します。

```
# チュートリアルその 6
<META>
title チュートリアル 6 (スキルレベル上限解放、結果表示の工夫、追加スキル検索)

<UI>
subsection 装備
width:150px
(0..1) フクなし:{フク}
(0..C) ポロシャツ:{フク} -> 1 防御力, 1 攻撃, 2 小スロ, 1 中スロ, 1 大スロ
(0..C) パーカー:{フク} -> 2 防御力, 2 小スロ
br
(0..1) クツなし:{クツ}
(0..C) サンドル:{クツ} -> 1 防御力
(0..C) スリッパ:{クツ} -> 2 防御力, 1 解放, 1 小スロ
br
(0..[0 1 2 3 4 5]) 攻撃珠:{珠} -> 1 攻撃, -1 小スロ
(0..[0 1 2]) 攻撃大珠:{珠} -> 2 攻撃, -1 大スロ
(0..[0 1 2 3]) 解放珠:{珠} -> 1 解放, -1 中スロ

subsection スキル条件・最低防御力指定
([0 1 2 3 4 5]..5)* 攻撃:{スキル}
([0 1 2 3]..3)* 解放:{スキル}
(0?..) 防御力

<RELATION>
{フク}.sum = {クツ}.sum = 1

大スロ >= 0
大スロ + 中スロ >= 0
大スロ + 中スロ + 小スロ >= 0

小以上空きスロ = 大スロ + 中スロ + 小スロ
中以上空きスロ <= 小以上空きスロ
中以上空きスロ <= 大スロ + 中スロ
大以上空きスロ <= 中以上空きスロ
大以上空きスロ <= 大スロ

小空きスロ = 小以上空きスロ - 中以上空きスロ
中空きスロ = 中以上空きスロ - 大以上空きスロ
大空きスロ = 大以上空きスロ

最大化 = 1000 防御力 + 小以上空きスロ

<UNLOCK>
解放 3, 攻撃 3 5
```



```
<QUERY>
query 検索 検索中止 追加検索 最大化
```

```
<SUMMARY>
width:50px
v 防御力
width:150px
n* {フク}
n* {クツ}
```

```
<DETAILS>
newcolumn 検索対象スキル
nv* {スキル}
```

```
newcolumn !対象外スキル
```

```
newcolumn 珠
nv* {珠}
```

```
newcolumn 空きスロット
nv 小空きスロ
nv 中空きスロ
nv 大空きスロ
```

```
newcolumn 所要時間
time
```

```
newcolumn 追加スキル
more 追加スキル {スキル}
```

7.1 入力ファイルの内容の説明

■UI セクション

```
<UI>
:
中略
:
(0..[0 1 2 3]) 解放珠:{珠} -> 1 解放, -1 中スロ

subsection スキル条件・最低防御力指定
([0 1 2 3 4 5]..5)* 攻撃:{スキル}
([0 1 2 3]..3)* 解放:{スキル}
(0?..) 防御力
```

解放珠を新設します。

また、「([0 1 2 3 4 5]..5)* 攻撃」と、スキルレベル指定のうしろに「*」を付けると、検索結果の表示で違いがあります。「*」が付いていて、かつ、そのスキルを検索してはいなかったとき、つまり、上の攻撃スキルの例だと、プルダウンで0を選択していたとき^{*4}、本来表示すべき列の次の列に表示します。

^{*4} 正確には、最小値、最大値がプルダウンメニュー等で設定できる場合、最小値は最も小さく、最大値は最も大きく設定されている場合が、「検索していない」状態としています。

これにより、検索対象スキルとおまけで付いてきたスキルを区別して検索結果に表示することができます。

■RELATION セクション

```
<RELATION>
{フク}.sum = {クツ}.sum = 1

大スロ >= 0
大スロ + 中スロ >= 0
大スロ + 中スロ + 小スロ >= 0

小以上空きスロ = 大スロ + 中スロ + 小スロ
中以上空きスロ <= 小以上空きスロ
中以上空きスロ <= 大スロ + 中スロ
大以上空きスロ <= 中以上空きスロ
大以上空きスロ <= 大スロ

小空きスロ = 小以上空きスロ - 中以上空きスロ
中空きスロ = 中以上空きスロ - 大以上空きスロ
大空きスロ = 大以上空きスロ

最大化 = 1000 防御力 + 小以上空きスロ

<UNLOCK>
解放 3, 攻撃 3 5
```

空きスロットを求めることを説明しますが少し複雑で、詳細は <https://github.com/13-480/lp-doc/blob/main/lpsim-v4.pdf> に説明があります。上の式で定義される「小空きスロ」「中空きスロ」「大空きスロ」は、そのまま空きスロット数を表すのではなく、上の式による制約の下、とりうる最大の値が空きスロット数になります。

そこで、「最大化 = 1000 防御力 + 小以上空きスロ」によって、「小空きスロ」「中空きスロ」「大空きスロ」も可能な限り大きくなるように、検索条件を指定します。このとき、もともと最大にしようとしていた防御力を 1000 倍して、1000 を越えることはない「小以上空きスロ」に影響されないようにして、防御力が等しい検索結果同士では、「小以上空きスロ」が大きい方を選ぶようにしています。

■UNLOCK セクション

```
<UNLOCK>
解放 3, 攻撃 3 5
```

上限解放は上のように記述します。この例では、解放スキルが 3 未満ならば攻撃スキルの上限は 3、解放スキルが 3 以上ならば攻撃スキルの上限は 5 と指定しています。

この指定を用いると、攻撃スキルが 3 より高いものを検索したい時に、解放スキルを 3 にする設定を手動でしていなくとも、必ず解放スキルが 3 以上になる結果のみが得られます。

■DETAILS セクション

```
<DETAILS>
newcolumn 検索対象スキル
```

```

nv* {スキル}

newcolumn !対象外スキル

newcolumn 珠
nv* {珠}

newcolumn 空きスロット
nv 小空きスロ
nv 中空きスロ
nv 大空きスロ


newcolumn 所要時間
time

newcolumn 追加スキル
more 追加スキル {スキル}

```

検索対象スキルの列の次に対象外スキルの列を追加しておきます。中身は空ですが、UI セクションで「*」指定があれば、この列に表示されることになります。

また、「newcolumn !対象外スキル」のように、文字列先頭に「!」がある場合は、既に前の列から移動してきたものがあったとしても、1 行目に表示します。



「time」はこの結果の検索に要した時間を表示します。

「more 追加スキル {スキル}」は、追加スキル検索のボタンを配置します。これをクリックすると、この検索結果の見出し部分で「n*」と指定され、表示されている装備を固定したときの、グループ「{スキル}」に登録されているものの最大値を検索し、表示します。

チュートリアル6 (スキルレベル上限解放、結果表示の工夫、追加スキル検索)

▼ 装備

☒ ポロシャツ ☒ パーカー
☒ サンダル ☒ スリッパ

3 攻撃珠 2 攻撃大珠 3 解放珠

▼ スキル条件・最低防御力指定

3 攻撃 0 解放 0 防御力

追加検索 検索結果の全消去

▼ 3 ポロシャツ スリッパ

検索対象スキル	対象外スキル	珠	空きスロット	所要時間	追加スキル
攻撃 3	解放 1	攻撃大珠 1	小空きスロ 3 中空きスロ 1 大空きスロ 0	0.065sec	追加スキル 攻撃Lv4 解放Lv3

▼ 4 パーカー スリッパ

上の例だと、ポロシャツ、スリッパを固定したとき、珠の付け変えて攻撃や解放スキルが最大いくらにできるかを検索した結果が表示されています。

8 雑多な補足

8.1 ワンセット防具

例えば、ポロシャツとサンダルはばらばらに装備することはできず、常に同時に装備することしかできない場合 (モンスターハンターワールドのワンセット防具に相当)、RELATION セクションに

```
<RELATION>
ポロシャツ = サンダル
```

と記述するとよいです (同時に 0 か、同時に 1 の可能性しかなくなるので)。装備の部位が 2 つより大きい場合も、「ポロシャツ = サンダル = メガネ」のようにするとよいです。

8.2 検索結果が少ないこと (線形計画法)

このツールで生成されるシミュレータの探索部分には、線形計画法が使われています。ゲームが異なってもシミュレータには共通する部分がありますが、高速化については個別の工夫がなされていました。これを、線形計画法のライブラリに丸投げすることで、検索と高速化はすべてライブラリに任せることができるというメリットがあります。

よく見るシミュレータでは、条件に合致するものがある程度少なければ、すべてを検索結果として表示します。しかし、線形計画法のライブラリは、通常、条件に合致するもののうち、例えば防御力最大のものを 1 つ結果として返します。つまり、すべてを結果として表示しません。このツールで生成されたシミュレータで検

索を実行したとき、1 つしか検索結果が表示されない^{*5}のはこれが理由です。

すべての結果を得たいときは、既に見付かった装備の組合せを除外して、もう一度検索をして、1 つずつ見付けばよいです。これが、「追加検索」ボタンの役割です。

8.3 追加検索の条件

その「追加検索」ボタンの動作の詳細を説明します。おおまかに言えば、検索結果欄に残っているものを除外して、検索実行するボタンです。

もう少し詳しく言うと、SUMMARY セクションで「n*」と指定されている装備の組合せは除外して検索します。

ただし、正しく機能するためには、「n*」と指定されている装備の所持数の上限が 1 である必要があります。また、「フクなし」のような、その部位の装備がない場合の、架空の装備を用意した方が適切な除外が行われるのでお勧めします。

8.4 パラメータ保存のタイミング

プルダウンメニュー、チェックボックス、テキストボックスの内容は保存され、リロードしたときや、ブラウザを閉じて再び開いたときに回復されます。保存のタイミングは、検索ボタンや検索結果の全消去ボタンをクリックしたときです。

9 おわりに

どうか身バレしませんように。そして、5ch スキルシミュレータ開発スレのみなさま、特に、MHRise 装備データ入力用の各ファイルのメンテナンスを下さっているみなさまに感謝いたします。

1 版. 2021 年 4 月 1 日

(2021 年 4 月 2 日 雑多な補足に追記)

2 版. 2021 年 5 月 11 日 独自言語の文法的大幅な変更。空きスロット数の表示方法。追加スキル検索。

^{*5} 実際には、既存のライブラリ「glpk.js」(<https://github.com/hgourvest/glpk.js>) に手を入れて、最良のものを発見する過程で見付かった、最良ではない結果も表示できるようにしているので、場合によっては 2 つ以上の結果が表示されることがあります。

従って、1 回目の検索で A、2 回目で B と C、3 回目で D が見付き、検索結果エリアに A、B、C、D の順に並んだときでも、防御力の大きい順とは限らないです。A \geq B \geq D と B \geq C は保証されますが、C と D の大小は不明です。